

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

TABLE OF CONTENTS

1.	SALES CUBE SHEET	1
	VOLUMES - LBS.....	2
	<i>Volume – Lbs, FY2009</i>	<i>2</i>
	<i>Volume – Lbs, FY2010</i>	<i>2</i>
	Administration of Actual + Fcst Account.....	2
	Adjusting DP Lbs	3
	Original Budget 2010	4
	<i>Volume – Lbs, FY2011</i>	<i>4</i>
	VOLUMES - CASES	5
	<i>Volume – Cases, FY2009.....</i>	<i>5</i>
	<i>Volume – Cases, FY2010.....</i>	<i>5</i>
	Adjusting DP Cases	6
	<i>Volume – Cases, FY2011</i>	<i>7</i>
	PRICES	7
	<i>Prices – FY2009</i>	<i>7</i>
	<i>Prices – FY2010</i>	<i>7</i>
	<i>Prices – FY2011</i>	<i>8</i>
	REVENUE.....	9
	STATISTICS.....	9
2.	REPORTS.....	10
	BUDGET PRESENTATION	10
	BUDGET ROLLUP.....	10

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

1. Sales Cube Sheet

→ **Note:** In this sheet, years of data are stacked on top of one another, like this:

FY2011 Lbs
FY2010 Lbs
FY2009 Lbs

Data has been forecasted in 12 time periods, April 2010 through March 2011. **The years on these time period columns should be ignored**, because multiple years of data are displayed in each month. (The headers on time period columns cannot be changed.)

Each user should go to sheet properties, and set the sheet to display FY2011 months.



→ **Note:** The quarter and year time periods do not need to be shown. Instead, cumulative accounts sum quantities and revenue to display YTD figures in each month. More on this, below.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

Volumes - Lbs

Volume – Lbs, FY2009

Actual Lbs from IFS are imported. The row called **Lbs Actual Cumulative FY2009** displays the YTD Lbs in each time period, with the following formula:

```
=if(fiscalmonth(this)=1,ACCT.Sales_Cube.Lbs_Actual_FY2009,  
ACCT.Sales_Cube.Lbs_Actual_FY2009  
+ACCT.Sales_Cube.Lbs_Act_Cum_FY2009[time=THIS-1])
```

→ **Administration Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department is added to the model structure.

Volume – Lbs, FY2010

Actual Lbs are imported from IFS. For the remaining time periods in this fiscal year, forecasted Lbs from DP are imported. (These two rows are hidden from view.)

The row called **Lbs Actual + Unadjusted DP FY2010** displays volumes for the entire year, with the following formula:


```
=if(ACCT.Sales_Cube.Trigger_FY2010  
[plan=Corporate Plan, rollup=yes, Product Family=All Product Families/SKUs]=0,  
ACCT.Sales_Cube.Lbs_Actual_FY2010,ACCT.Sales_Cube.Lbs_DP_FY2010)
```

→ **Note:** DP data will not be deleted as actual data is imported in subsequent months. This forecasted DP data will stay in those time periods, for reference.

Administration of Actual + Fcst Account

The Actual + Fcst rows have formulas referencing the actual data or the forecasted data, depending on the time period. There is a “trigger” account that uses a value of 0 or 1 to tell the system whether the time period is an actual or a forecasted time period.

Go to Plan = Corporate Plan (Only) and Product Family = All Product Families/SKUs (Uncategorized), and **enter a zero or a one in these trigger accounts.** Be sure to update this each month after importing actual data.



WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

In actual months, the values should be zero. In forecast months, the values should be one.

Actual or Plan Time Period Triggers				
Trigger FY2011	1	1	1	1
Trigger FY2010				

→ **Note:** The cube sheet has been restricted so that these “trigger” accounts are not available to any plans. The administrator should unrestrict these accounts once each month, when they need to be updated, and then restrict them again.

Adjusting DP Lbs

DP Forecasted Units can be adjusted three ways, in the order described here.

First, manual adjustments can be made to the DP Lbs forecast for the last three months (Jan, Feb, Mar) of FY2010. **Enter positive or negative numbers, representing lbs.** These lbs are added to the lbs imported from DP.

→ **Note:** The resulting percentage change is calculated and displayed in the row called **Lbs DP FY2010 Manual % Change**. This row is hidden from view, but has the following formula:

$$\text{if}((\text{div}(\text{ACCT.Sales_Cube.Lbs_DP_Adj_Total_FY2010}, \text{ACCT.Sales_Cube.Lbs_DP_FY2010}))=0, 0, (\text{div}(\text{ACCT.Sales_Cube.Lbs_DP_Adj_Total_FY2010}, \text{ACCT.Sales_Cube.Lbs_DP_FY2010}))-1)$$

→ **Note:** The percentage change calculated here will be applied to the DP units in the same time periods in FY11. More on this, below.

The account called **Lbs DP Adjusted Total FY2010** adds the DP manual adjustment to the imported DP value. This account is hidden from view, but has the following formula:

$$\text{ACCT.Sales_Cube.Lbs_DP_FY2010} + \text{ACCT.Sales_Cube.Lbs_DP_FY2010_Enter}$$

Secondly, the user can enter a volume percent change adjustment to the DP pounds. The row called **Lbs Calc'd as DP * Vol Chg, FY2010** is hidden, but was designed to display the number of Lbs adjusted as a result of this entered percentage. It has the following formula:

$$\text{ACCT.Sales_Cube.Lbs_DP_Adj_Total_FY2010} \\ * (1 + \text{ACCT.Sales_Cube.Vol_Pct_Chg_FY2010_Enter})$$

Thirdly, the user can enter a positive or negative number of units in the row called **Lbs Forecasted Enter Adj FY2010**.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

The final adjusted Lbs number is displayed in the row called **Lbs Forecasted Adjusted Total FY2010**, which has the following formula:

=ACCT.Sales_Cube.Lbs_Calcd_DP_Vol_Chg_FY2010
+ACCT.Sales_Cube.Lbs_Fcst_Enter_FY2010

Following this is a row called **Lbs Actual + Adjusted DP FY2010**, which uses the same “trigger” as described above to place actual Lbs in actual months and adjusted DP units in forecasted months, so that Lbs for the total year can be seen. It has the following formula:

if(ACCT.Sales_Cube.Trigger_FY2010
[plan=Corporate Plan, rollup=yes, Product Family=All Product Families/SKUs]=0
,ACCT.Sales_Cube.Lbs_Actual_FY2010,ACCT.Sales_Cube.Lbs_Adj_Fcst_Total_FY2010)

Following this is a row called **Lbs Actual + Forecasted Cumulative FY2010**, which displays volumes for the entire year, with the following formula:

=if(fiscalmonth(this)=1,ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010,
ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010+
ACCT.Sales_Cube.Lbs_Act_plus_Fcst_Cum_FY2010[time=THIS-1])

→ **Administration Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department is added to the model structure.

Original Budget 2010

Volume in Lbs from the Original Budget for FY2010 is displayed at the end of this section, for reference. Budget values have been imported at the Consumer/National/Institutional plan level, and at the PF detail level. **Nine months of values have been imported into December, and the total year has been imported into March.** The account is cumulative/plan by balance so that the last value in March is also displayed in Q4 and in the year total columns.

The statistic called Lbs Fcst div by Orig should be calculated in Excel report templates, instead of here on the sheet. Since the original budget data resides at a different level than the forecast data, the comparison doesn't make sense in calculated cube accounts.

Volume – Lbs, FY2011

This section works the same way as Volume – Lbs, FY2010, with the following exceptions:

At this point, the entire year is forecasted. There are not yet any actual time periods. As the fiscal year progresses, and the time periods become a combination of Actual + Fcst, the Actual and Actual + Forecasted rows will behave as described for FY2010.

The first adjustment to the DP Lbs is not entered, but driven from the adjustment made to the last three months of FY2010. The change made to the DP Lbs units for FY2010 drives a change in the same percentages to the Lbs units in the same months in FY2011. (E.g. if March 2010 Lbs units are adjusted by 20%, then March 2011 DP Lbs units will also be adjusted by 20%.)

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

Lbs DP Adjusted per Prior Year FY2011 are calculated with this formula:

if(ACCT.Sales_Cube.Lbs_DP_FY2011=0,
ACCT.Sales_Cube.Lbs_Adj_Fcst_Total_FY2010,ACCT.Sales_Cube.Lbs_DP_FY2011
*if(div(ACCT.Sales_Cube.Lbs_Adj_Fcst_Total_FY2010,ACCT.Sales_Cube.Lbs_DP_FY2010)=0,
1,div(ACCT.Sales_Cube.Lbs_Adj_Fcst_Total_FY2010,ACCT.Sales_Cube.Lbs_DP_FY2010)))

In other words, if the DP starting point is zero, then take the amount entered into Lbs adjustment for the same period in FY2010, otherwise, divide the adjusted DP Lbs for 2010 by the unadjusted DP Lbs for 2010. Multiply this percentage by the unadjusted DP Lbs for FY2011. (If there is no change between the adjusted and unadjusted Lbs in FY2010, multiply the unadjusted DP Lbs for FY2011 by 1.)

The difference between FY2011 DP Lbs as imported and FY2011 DP Lbs as calculated with this percentage change can be seen (as number of Lbs) in the row called **+/- Lbs DP Calc'd Adj FY2011**, which is hidden from view but has the following formula:

ACCT.Sales_Cube.Lbs_DP_Adj_per_Prior_Yr_FY2011-ACCT.Sales_Cube.Lbs_DP_FY2011

Volumes - Cases

The Volume – Cases section is very similar to the Volume – Lbs section, but it is simpler.

Volume – Cases, FY2009

Actual cases from IFS are imported. The row called **Cases Actual Cumulative FY2009** displays the YTD Cases in each time period, with the following formula:

=if(fiscalmonth(this)=1,ACCT.Sales_Cube.Cases_Actual_FY2009,
ACCT.Sales_Cube.Cases_Actual_FY2009
+ACCT.Sales_Cube.Cases_Actual_Cum_FY2009[time=THIS-1])

→ **Administration Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department is added to the model structure.

Volume – Cases, FY2010

Actual cases are imported from IFS. For the remaining time periods in this fiscal year, forecasted cases from DP are imported. The row called **Cases Actual + Unadjusted DP FY2010** displays volumes for the entire year, with the following formula:

=if(ACCT.Sales_Cube.Trigger_FY2010
[plan=Corporate Plan, rollout=yes, Product Family=All Product Families/SKUs]=0,
ACCT.Sales_Cube.Lbs_Actual_FY2010,ACCT.Sales_Cube.Lbs_Forecasted_Adj_Total_FY2010)

→ **Note:** DP data will not be deleted as actual data is imported in subsequent months. This forecasted DP data will stay in those time periods, for reference.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

Adjusting DP Cases

DP Forecasted Cases are adjusted in three ways, in the order described here.

First, in the row called **Cases DP Adjusted per Prior Year FY2010**, a calculated adjustment is made to the DP Cases forecast for the last three months (Jan, Feb, Mar) of FY2010. This calculation is based on the manual change made to DP Lbs in these same three months. The resulting percentage change is applied here to DP cases.

The number of cases in the adjustment itself is calculated and displayed on the row called +/- Cases DP Calc'd Adj FY2010.

Secondly, the volume percentage change that is entered in the FY2010 Lbs section above is applied to the adjusted DP cases, in the row called **Cases Calc'd as DP * Vol Chg, FY2010**, with the following formula:

$$\text{ACCT.Sales_Cube.Cases_DP_Adj_per_Prior_Yr_FY2010}^* \\ (1 + \text{ACCT.Sales_Cube.Vol_Pct_Chg_FY2010_Enter})$$

Thirdly, the user can enter a positive or negative number of cases in the row called **Cases Forecasted Enter Adj FY2010**.

The final adjusted Cases number is displayed in the row called **Cases Forecasted Adjusted Total FY2010**, which has the following formula:

$$\text{ACCT.Sales_Cube.Cases_Calcd_DP_Vol_Chg_FY2010} \\ + \text{ACCT.Sales_Cube.Cases_Fcst_Enter_FY2010}$$

Following this is a row called **Cases Actual + Adjusted Forecasted FY2010**, which uses the same "trigger" as described above to place actual Cases in actual months and adjusted DP units in forecasted months, so that Cases for the total year can be seen. It has the following formula:

$$\text{if}(\text{ACCT.Sales_Cube.Trigger_FY2010} \\ [\text{plan=Corporate Plan, Product Family=All Product Families/SKUs}]=0, \\ \text{ACCT.Sales_Cube.Cases_Actual_FY2010}, \text{ACCT.Sales_Cube.Cases_Adj_Fcst_Total_FY2010})$$

Following this is a row called **Cases Actual + Forecasted Cumulative FY2010**, which displays volumes for the entire year, with the following formula:

$$\text{=if}(\text{fiscalmonth}(\text{this})=1, \text{ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010}, \\ \text{ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010} + \\ \text{ACCT.Sales_Cube.Lbs_Act_plus_Fcst_Cum_FY2010}[\text{time=THIS-1}])$$

→ **Administration Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department is added to the model structure.

Volume in Lbs from the Original Budget for FY2010 is displayed at the end of this section, for reference.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

Volume – Cases, FY2011

This section works the same way as Volume – Cases, FY2010, with the following exceptions:

At this point, the entire year is forecasted. There are not yet any actual time periods. As the fiscal year progresses, and the time periods become a combination of Actual + Fcst, the Actual and Actual + Forecasted rows will behave as described for FY2010.

The first adjustment to the DP cases is driven from the adjustment made to the last three months of FY2010. The change made to the DP cases for FY2010 drives a change in the same percentages to the cases in the same months in FY2011. (E.g. if March 2010 cases are adjusted by 20%, then March 2011 DP cases will also be adjusted by 20%.)

Prices

Prices – FY2009

Actual prices have been imported from IFS.

Prices – FY2010

Actual prices have been imported from IFS.

In the account called **ASP Calcd from Prior Year FY2010**, forecasted prices for the last three months of FY2010 are calculated as (the last three months of revenue divided by the last three months of Lbs), with the following formula:

=div(ACCT.Sales_Cube.Rev_Actual_FY2010[time=10/2010:12/2010]
,ACCT.Sales_Cube.Lbs_Actual_FY2010[time=10/2010:12/2010])

→ **Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department are added to the model structure.

→ **Note:** Accounts in Adaptive have the following choices for time period rollups (“sideways” rollups) and plan/dimension rollups (“up and down” rollups):

- Sum
- Simple Average
- Weighted Average (you choose another account to use as the weight)
- Display text instead of values

Most of the ASP accounts roll up as weighted averages, with the denominator account as the weight, but this is not entirely accurate. This will not produce the same result as if revenue were divided by Lbs for the total time period rollup. The best workaround is to create the accurate calculation in Excel, in report templates.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

Following this calculated price, there are two opportunities for manual adjustment, by entering either a positive or negative price value, and/or a positive or negative percentage.



ASP FY2010	
ASP Actual FY2010	2.4551
ASP Calcd from Prior Year FY2010	2.0488
ASP Planned Enter Adj FY2010	1.0000
ASP Planned Enter % Adj FY2010	-2.00%

The **ASP Adjusted Forecasted Total YF2010** (which is used to drive revenue,) is calculated with the following formula:

$$\begin{aligned} &(\text{ACCT.Sales_Cube.ASP_Calcd_from_Prior_Yr_FY2010} \\ &+ \text{ACCT.Sales_Cube.ASP_Forecasted_Enter_Adj_FY2010}) \\ &*(1 + \text{ACCT.Sales_Cube.ASP_Forecasted_Enter_Pct_Adj_FY2010}) \end{aligned}$$

The account called **ASP Actual + Forecasted FY2010** displays the price for the entire year, with the following formula:

$$\begin{aligned} &= \text{if}(\text{ACCT.Sales_Cube.Trigger_FY2010} \\ &[\text{plan}=\text{Corporate Plan, Product Family}=\text{All Product Families/SKUs}]=0, \\ &\text{ACCT.Sales_Cube.ASP_Actual_FY2010, ACCT.Sales_Cube.ASP_Forecasted_Total_FY2010}) \end{aligned}$$

Prices – FY2011

Actual prices have been imported from IFS.

In the account called **ASP Calcd from Prior Year FY2011**, the end-of-year FY2010 adjusted price is carried into the price in all the months of FY2011, with the following formula:

$$= \text{ACCT.Sales_Cube.ASP_Adj_Forecasted_Total_FY2010} [\text{time}=03/2011]$$

→ **Administration Note:** Due to restrictions on formula writing in cube sheet calculated accounts, **this formula must be entered on the sheet for all Product Families in all Departments.** After initial implementation, the formula should be added by an administrator whenever a new product family or department is added to the model structure.

In the same way as in the FY2010 prices, following this calculated FY2011 price, there are two opportunities for manual adjustment, by entering either a positive or negative price value, and/or a positive or negative percentage.

The **ASP Adjusted Forecasted Total FY2011** (which is used to drive revenue,) is calculated with the following formula:

$$\begin{aligned} &(\text{ACCT.Sales_Cube.ASP_Calcd_from_Prior_Yr_FY2011} \\ &+ \text{ACCT.Sales_Cube.ASP_Forecasted_Enter_Adj_FY2011}) \\ &*(1 + \text{ACCT.Sales_Cube.ASP_Forecasted_Enter_Pct_Adj_FY2011}) \end{aligned}$$

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

The row called **ASP Actual + Forecasted FY2011** will be useful when FY2011 contains some months of actual and some months of forecast. It has the following formula:

if(ACCT.Sales_Cube.Trigger_FY2011
[plan=Corporate Plan, Product Family=All Product
Families/SKUs]=0,ACCT.Sales_Cube.ASP_Actual_FY2011,
ACCT.Sales_Cube.ASP_Adj_Forecasted_Total_FY2011)

There is a statistic at the end of this section called **ASP Pct Chg Result from Prior Yr FY2011**. It has the following formula:

div(ACCT.Sales_Cube.ASP_Act_plus_Forecasted_FY2011,
ACCT.Sales_Cube.ASP_Act_plus_Forecasted_FY2010)

In other words, divide the adjusted forecasted ASP for this year by the Actual or adjusted ASP (depending on the time period) from last year.

Revenue

FY2009 contains imported actuals and calculated Cumulative actuals.

FY2010 and FY2010 are the same, as described here:

Actual revenue is imported from IFS.

Revenue Forecasted (Calc'd) Unadjusted is calculated with formulas like this:

ACCT.Sales_Cube.Lbs_Adj_Fcst_Total_FY2011
*ACCT.Sales_Cube.ASP_Adj_Forecasted_Total_FY2011

There is a row for manual adjustments. **Revenue Forecasted Adjusted Total** adds the adjustment to the unadjusted calculated revenue.

Revenue Actual + Forecasted accounts rely on the "trigger" described above to place actual or forecasted data in each time period.

if(ACCT.Sales_Cube.Trigger_FY2010
[plan=Corporate Plan, Product Family=All Product Families/SKUs]=0,
ACCT.Sales_Cube.Rev_Actual_FY2010,ACCT.Sales_Cube.Revenue_Fcst_Adj_Total_FY2010)

Revenue Actual + Adjusted Forecasted Cumulative accounts calculate the YTD revenue in each time period, as described above.

Statistics

There is a statistics section at the bottom of the sheet. The calculations presented here are repeated from (nearly) identical rows up above, in the volume, price, and revenue sections.

→ **Note:** These statistics accounts are all calculated percentages. They make sense at the individual subplan/sub Product Family level, but they are not completely accurate as they roll up the Plan or PF tree. Their plan/PF rollups are calculated as a weighted average. For completely accurate numbers, calculations in Excel report templates are recommended.

WNA
Adaptive Planning Implementation Notes
Prepared by Tracy Schreiber
January 2010

There are two statistics for each of FY2010 and FY2011, for each of volume, price, and revenue. For example, in volume for FY2010:

Vol % Chg Result from Prior Yr FY2010 has the following formula:

$$\text{if}((\text{div}(\text{ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010}, \\ \text{ACCT.Sales_Cube.Lbs_Actual_FY2009})))=0,0, \\ (\text{div}(\text{ACCT.Sales_Cube.Lbs_Actual_Plus_Adj_DP_FY2010},\text{ACCT.Sales_Cube.Lbs_Actual_FY2009}))-1)$$

Vol Pct Chg FYTD Resulting from Prior Yr FY2010 has the following formula:

$$\text{if}((\text{div}(\text{ACCT.Sales_Cube.Lbs_Act_plus_Fcst_Cum_FY2010}, \\ \text{ACCT.Sales_Cube.Lbs_Act_Cum_FY2009})))=0,0, \\ (\text{div}(\text{ACCT.Sales_Cube.Lbs_Act_plus_Fcst_Cum_FY2010}, \\ \text{ACCT.Sales_Cube.Lbs_Act_Cum_FY2009}))-1)$$

2. Reports

The following shared reports are sent to Excel as formatted Template reports:

Budget Presentation

This report displays data for the entire company. It has Product Family rollups in the rows axis. It also has specific product families within each family, as rows highlighted in blue font. Cube accounts such as Revenue Actual FY2009 are in the columns axis.

→ **Note:** This report template will require some row maintenance if the Product Family dimension tree changes.

Budget Rollup

There are two versions of this report. One displays data for the entire company. The other has the Institutional Sales Group (PF rollup) in the Filter.

This report has the entire Product Family tree in its rows axis. Cube accounts such as Revenue Actual FY2009 are in the columns axis. There are three time elements in the columns axis: YTD as of December, Total Fiscal Year, and Q4.

→ **Note:** This report template will require some row maintenance if the Product Family dimension tree changes.